# NAG Library Routine Document

# G05SPF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

G05SPF generates a vector of pseudorandom numbers from a triangular distribution with parameters $x_{\text{min}}$, $x_{\text{med}}$ and $x_{\text{max}}$.

## 2 Specification

```
SUBROUTINE G05SPF(N, XMIN, XMED, XMAX, STATE, X, IFAIL)
INTEGER           N, STATE(lstate), IFAIL
double precision  XMIN, XMED, XMAX, X(N)
```

## 3 Description

The triangular distribution has a PDF (probability density function) that is triangular in profile. The base of the triangle ranges from $x = x_{\text{min}}$ to $x = x_{\text{max}}$ and the PDF has a maximum value of $\dfrac{2}{x_{\text{max}} - x_{\text{min}}}$ at $x = x_{\text{med}}$. If $x_{\text{min}} = x_{\text{med}} = x_{\text{max}}$ then $x = x_{\text{med}}$ with probability 1; otherwise the triangular distribution has PDF:

$$f(x) = \frac{x - x_{\text{min}}}{x_{\text{med}} - x_{\text{min}}} \times \frac{2}{x_{\text{max}} - x_{\text{min}}} \quad \text{if } x_{\text{min}} \leq x \leq x_{\text{med}},$$

$$f(x) = \frac{x_{\text{max}} - x}{x_{\text{max}} - x_{\text{med}}} \times \frac{2}{x_{\text{max}} - x_{\text{min}}} \quad \text{if } x_{\text{med}} < x \leq x_{\text{max}},$$

$$f(x) = 0 \quad \text{otherwise.}$$

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SPF.

## 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

## 5 Parameters

1: N – INTEGER *Input*

*On entry*: $n$, the number of pseudorandom numbers to be generated.

*Constraint*: N $\geq 0$.

2: XMIN – ***double precision*** *Input*

*On entry*: the end point $x_{\text{min}}$ of the triangular distribution.

3: XMED – ***double precision*** *Input*

*On entry*: the median of the distribution $x_{\text{med}}$ (also the location of the vertex of the triangular distribution at which the PDF reaches a maximum).

*Constraint*: XMED $\geq$ XMIN.

4: XMAX – ***double precision*** *Input*

*On entry*: the end point $x_{\text{max}}$ of the triangular distribution.

*Constraint*: XMAX $\geq$ XMED.

5: STATE($lstate$) – INTEGER array *Communication Array*

**Note**: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

*On entry*: contains information on the selected base generator and its current state.

*On exit*: contains updated information on the state of the generator.

6: X(N) – ***double precision*** array *Output*

*On exit*: the $n$ pseudorandom numbers from the specified triangular distribution.

7: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to 0, $-1$ or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit*: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value $-1$ or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $< 0$.

IFAIL $= 3$

On entry, XMED $<$ XMIN.

IFAIL $= 4$

On entry, XMAX $<$ XMED.

IFAIL $= 5$

On entry, STATE vector was not initialized or has been corrupted.

## 7 Accuracy

Not applicable.

## 8 Further Comments

None.

## 9 Example

This example prints five pseudorandom numbers from a triangular distribution with parameters $x_{\min} = -1.0$, $x_{\text{med}} = 0.5$ and $x_{\max} = 1.0$, generated by a single call to G05SPF, after initialization by G05KFF.

### 9.1 Program Text

```
*     G05SPF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*     .. Local Scalars ..
      DOUBLE PRECISION XMAX, XMED, XMIN
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05SPF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05SPF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameters XMIN,XMAX and XMED
      XMIN = -1.0D0
      XMAX = 1.0D0
      XMED = 0.5D0
*     Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*     Generate the variates
      IFAIL = 1
      CALL G05SPF(N,XMIN,XMED,XMAX,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*     Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
   20 CONTINUE
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05SPF returned with IFAIL = ',I5)
      END
```

### 9.2 Program Data

None.

## 9.3 Program Results

```
 G05SPF Example Program Results

     0.3817
    -0.4348
     0.4960
     0.5509
    -0.4398
```